

```

10 'CONSUMMATION DE CARBURANT
20 'Michel ARCHAMBAULT--1985 / AMSTRAD
40 DEFINT I,J,M,N
50 GOSUB 51000:'AZERTY
60 INK 0,1:INK 1,24
70 PAPER 2:PEN 3:BORDER 15
100 'TITRE
110 MODE 0:CLS
120 LOCATE 5,4:PRINT "C O N S O M"
130 FOR I=0 TO 200 STEP 2:PLOT 0,I,9
140 DRAWR 640,0:NEXT:PEN 0
150 FOR I=1 TO 20:LOCATE I,7:PRINT "$":F
OR J=1 TO 200:NEXT:NEXT
160 FOR I=1 TO 1000:NEXT
170 FOR Z=200 TO 15 STEP -0.5:PLOT 0,Z,2
180 DRAWR 640,0:NEXT:LOCATE 2,23:PRINT "
M.Archambault-1985"
190 FOR I=1 TO 2000:NEXT
700 'VOITURE
710 MODE 1:CLS:PEN 3
720 LOCATE 12,5:PRINT "NOM DE LA VOITURE
:":PEN 0
730 LOCATE 7,7:PRINT "(sans espace ni po

```

```

nctuation)"
740 LOCATE 8,9:PRINT "(commencez par une
lettre)"
750 LOCATE 17,12:PRINT STRING$(8,"."):LO
CATE 17,12:INPUT "",VOIT$
760 IF VAL(VOIT$)>0 OR LEN(VOIT$)>8 THEN
PRINT CHR$(7):GOTO 750
1000 ' MENU
1010 PEN 3:PAPER 2:CLS:FPP=0
1020 LOCATE 17,3:PRINT "M E N U":PEN 0
1030 LOCATE 12,6:PRINT "C : CHARGEMENT F
ICHER"
1040 LOCATE 12,8:PRINT "E : ENTREE DES L
ITRES."
1050 LOCATE 12,10:PRINT "S : SAUVEGARDE
FICHER."
1060 LOCATE 12,12:PRINT "P : PREMIERE FO
IS."
1070 LOCATE 12,14:PRINT "F : FIN."
1080 TEX$="CESPF":PEN 3:GOSUB 50000:PEN
0
1090 ON K GOSUB 2000,5000,10000,15000,20
000

```

```

1100 GOTO 1000
2000 'CHARGEMENT DU FICHIER
2010 PAPER 0:PEN 3:BORDER 2:CLS
2020 LOCATE 6,6:PRINT "CHARGEMENT DU FIC
HIER VOITURE.":PEN 1
2030 LOCATE 19,12:PRINT "OK ?":TEX$="ON"
:GOSUB 50000:LOCATE 1,24:PRINT CHR$(18)
2040 IF K=2 THEN 2900
2050 LOCATE 4,15:PRINT "Préparez la Cass
ette ou Disquette."
2060 LOCATE 9,17:PRINT "puis pressez une
Touche...":CALL &BB06
2100 OPENIN VOIT$
2110 INPUT #9,DAT0$,DATD$,DATF$,K0,KD,DS
P,KF,LS,LTOT
2120 CLOSEIN
2200 CLS
2210 LOCATE 12,4:PEN 3:PRINT "VOITURE ";
VOIT$;" :":PEN 2
2220 LOCATE 12,7:PRINT "Précédente Saisi
e:"
2230 LOCATE 10,10:PRINT "du ";DATD$;" au
";DATF$
2240 LOCATE 10,12:PRINT "de";KD;"km à";K
F;"km"
2250 LOCATE 19,16:PEN 1:PRINT "OK ?":TEX
$="ON":GOSUB 50000
2260 IF K=2 THEN 2900
2900 RETURN
5000 'ENTREE DES DONNEES
5010 PEN 3:CLS:IF FPF=1 THEN 5200
5050 LOCATE 12,3:PRINT "ENTREE DES DONNE
ES:":PEN 0
5060 LOCATE 4,6:PRINT "Dernier Relevé: ";
DATF$;KF;"km"
5070 LOCATE 15,8:PRINT "Date Départ:":GO
SUB 12000
5080 LOCATE 11,17:INPUT "km départ: ",KD
$:KD=VAL(KD$):IF KD<KF THEN 5080
5090 LOCATE 10,20:PRINT DAT$;" &";KD;"km
: OK ?":TEX$="ON":GOSUB 50000
5100 IF K=2 THEN 5000
5110 DAT1$=DAT$:KMP=KF
5200 'TABLEAU DE SAISIE
5210 WINDOW #1,1,40,1,3:PAPER #1,1:PEN #
1,0:CLS #1
5220 PRINT #1:PRINT #1,TAB(4);"Litres: ";
TAB(17);"PLEIN";TAB(28);"COMPTEUR:"
5230 PRINT #1,TAB(4);"(FIN=F)";TAB(18);"
(P)";TAB(30);"(km)"
5240 WINDOW #2,1,40,4,6:PAPER #2,1:PEN #
2,0:CLS #2
5250 WINDOW #3,1,19,7,11:PAPER #3,2:PEN
#3,0:CLS #3:PRINT #3
5260 PRINT #3,TAB(3);"Plein a Plein:":PR
INT #3
5270 PRINT #3,TAB(3);"L";TAB(10);"km";TA
B(15);"L/100"
5280 WINDOW #4,20,40,7,11:PAPER #4,0:PEN

```

```

#4,3:CLS #4:PRINT #4
5290 PRINT #4,TAB(5);"Bilan cumulé:":PRI
NT #4
5300 PRINT #4,TAB(4);"L";TAB(10);"km";TA
B(16);"L/100"
5310 WINDOW #5,1,19,11,25:PAPER #5,2:PEN
#5,0:CLS #5
5320 WINDOW #6,20,40,11,25:PAPER #6,0:PE
N #6,3:CLS #6
5330 BORDER 15
6000 'SAISIE
6005 N=1:M=1:LT=0:TOLT=0
6010 LOCATE #2,6,N:INPUT #2,"",L$:IF UPP
ER$(L$)="F" THEN 7000
6030 L=VAL(L$):IF L=0 THEN PRINT CHR$(7)
::GOTO 6010
6040 LT=LT+L
6050 LOCATE #2,19,2:INPUT #2,"",P$:IF P$
=" " THEN N=N+1:GOTO 6010
6055 IF UPPER$(P$)="E" THEN LT=LT-L:N=N-
1:GOTO 6010
6060 IF UPPER$(P$)<>"P" THEN 6050
6070 LOCATE #2,29,2:INPUT #2,"",KM$:KM=V
AL(KM$)
6080 IF KM<=KMP THEN PRINT CHR$(7)::GOTO
6070
6100 DKM=KM-KMP:KMP=KM:CONP=100*LT/DKM
6110 TOLT=TOLT+LT:TKM=TKM+DKM:CONG=100*T
OLT/TKM
6150 LOCATE #5,2,M:PRINT #5,USING"###.#";
LT
6160 LOCATE #5,8,M:PRINT #5,USING"####";
DKM
6170 LOCATE #5,15,M:PRINT #5,USING"###.#"
;CONP
6180 LOCATE #6,2,M:PRINT #6,USING"###.#"
;TOLT
6190 LOCATE #6,9,M:PRINT #6,USING"####";
TKM
6200 LOCATE #6,17,M:PRINT #6,USING"###.#"
;CONG
6210 M=M+1:N=N+1:LT=0:GOTO 6010
7000 'TABLEAU FINAL
7010 PAPER 0:CLS
7020 LOCATE 8,6:PEN 1:PRINT "Date de ce
dernier Plein :":PEN 2:GOSUB 12000
7030 CLS:BORDER 12:LOCATE 2,3:PRINT "A c
e Relevé du ";DAT1$;" au ";DAT$
7040 LOCATE 1,5:PRINT TKM;"km avec";ROUN
D(TOLT,1);"L,soit";ROUND(CONG,1);"L/100
km"
7050 PRINT STRING$(40,"_"):IF FPF=1 THEN
7090
7060 PEN 2:LOCATE 1,8:PRINT "Relevé préc
édent,du ";DATD$;" au ";DATF$
7070 LOCATE 1,10:PRINT DSP;"km avec";ROU
ND(LS,1);"L,soit";ROUND(100*LS/DSP,1);"L
/100 km"
7080 PRINT STRING$(40,"_")

```

```

7090 LS=TOLT:LTOT=LTOT+TOLT:DATD$=DAT1$:
DATF$=DAT$:KF=KM:DSP=TKM
7100 PEN 3:LOCATE 3,14:PRINT "BILAN GENE
RAL du ";DAT0$;" au ";DATF$
7110 LOCATE 1,16:PRINT KF-K0;"km avec";R
OUND(LTOT,1);"L,soit";ROUND(100*LTOT/(KF
-K0),1);"L/100 km"
7120 PEN 1:LOCATE 8,20:PRINT "SAUVEGARDE
DE CES VALEURS:"
7130 LOCATE 3,22:PRINT "Préparez la Cass
ette ou Disquette,"

7140 LOCATE 8,24:PRINT "puis pressez une
Touche..."
7150 CALL &BB06
7160 GOSUB 10000
7170 RETURN
10000 ' ENREGISTREMENT DU FICHIER
10010 OPENOUT VOIT$
10020 WRITE #9,DAT0$,DATD$,DATF$,K0,KD,D
SP,KF,LS,LTOT
10030 CLOSEOUT
10900 RETURN
12000 'ENTREE DATE
12010 LOCATE 17,10:INPUT "jour: ",J$:J=V
AL(J$):IF J<1 OR J>31 THEN 12010
12020 LOCATE 17,12:INPUT "mois: ",M$:M=V
AL(M$):IF M<1 OR M>12 THEN 12020
12030 LOCATE 16,14:INPUT "année: ",AN$:A
N$=RIGHT$(AN$,2):IF VAL(AN$)<80 THEN 120
30
12040 DAT$=J$+ "/" +M$+ "/" +AN$
12900 RETURN
15000 'PREMIERE FOIS
15010 BORDER 2:PAPER 1:PEN 0:CLS
15020 LOCATE 2,4:PRINT "Vous ouvrez un N
OUVEAU fichier voiture."
15030 LOCATE 18,6:PRINT "OK ?":TEX$="ON"
:GOSUB 50000:IF K=2 THEN 15900
15035 CLS:LOCATE 9,2:PRINT"Au tout premi
er Plein:"
15040 LOCATE 8,4:INPUT"Kilométrage Dépar
t:",K0$
15050 K0=VAL(K0$):IF K0=0 THEN 15040
15060 LOCATE 16,7:PRINT "en Date du:":GO
SUB 12000
15065 LOCATE 8,18:PRINT "N'ENTREZ PAS CE
S LITRES !":FOR I=1 TO 4000:NEXT
15070 DAT0$=DAT$:DATD$=DAT0$:DATF$=DAT0$
:DAT1$=DAT0$
15080 KF=K0:LS=0:LTOT=0:KMP=K0:KD=K0
15090 FPF=1
15100 PAPER 0:GOSUB 5000
15900 RETURN
20000 'FINAL
20010 PAPER 0:PEN 1:CLS
20020 LOCATE 10,12:PRINT "VOUS POUVEZ ET
EINDRE"

```

```

20030 LOCATE 11,14:PRINT "ou relancer pa
r RUN.":PRINT:PRINT
20100 END
49999 END
50000 'REPONSE A UN MENU
50010 LT=LEN(TEX$):R$=""
50020 LOCATE 15-LT,24:PRINT "Réponse (";
50030 FOR I=1 TO LT-1
50040 PRINT MID$(TEX$,I,1);",";:NEXT
50050 PRINT RIGHT$(TEX$,1);")";CHR$(154)
;CHR$(243);CHR$(207)
50060 TEX$=UPPER$(TEX$)
50070 WHILE R$="" :R$=INKEY$:WEND
50080 R$=UPPER$(R$):K=INSTR(TEX$,R$)
50090 IF K=0 THEN R$="":PRINT CHR$(7);:G
OTO 50070
50100 RETURN
51000 ' AZERTY ACCENTUE
51010 SYMBOL AFTER 6
51020 SYMBOL 64,96,48,120,12,124,204,118
,0
51030 SYMBOL 91,0,56,108,56,0,0,0,0
51040 SYMBOL 92,0,0,60,102,96,102,60,24
51050 SYMBOL 93,60,96,60,102,60,6,60,0
51060 SYMBOL 123,12,24,60,102,126,96,60,
0
51080 SYMBOL 125,48,24,60,102,126,96,60,
0
51100 RETURN
65000 -----LEGENDE DES VARIABLES-----
65001 CONG : CONSOMMATION GLOBALE
65002 CONP : CONSOM. DE PLEIN A PLEIN
65003 DAT$ : DATE AU CLAVIER
65004 DAT0$ : DATE OUVERTURE FICHIER
65005 DAT1$ : DATE DEPART EN COURS
65006 DATD$ : DATE DEPART PRECEDENT
65007 DATF$ : DATE FIN DE SAISIE
65008 DKM : DISTANCE ENTRE DEUX PLEINS
65009 DSP : DISTANCE SAISIE PRECEDENTE
65010 FPF : FLAG PREMIERE FOIS
65011 I : UNITE DE COMPTAGE
65012 K : INDICE DE MENU
65013 K0 : KM A L'OUVERTURE DU FICHIE
R
65014 KD : KM DEPART SAISIE
65015 KF : KM FIN DE SAISIE
65016 KM : KM AU PLEIN
65017 KMP : KM AU PLEIN PRECEDENT
65018 L : LITRES ENTRES
65019 LS : LITRES DE LA SAISIE
65020 LT : LITRES ENTRE DEUX PLEINS
65021 LTOT : LITRES TOTAUX
65022 M,N : COMPTAGES DE LIGNES
65023 TEX$ : SIGLES DU MENU
65024 TKM : DISTANCE TOTALE EN SAISIE
65025 TOTL : LITRES SAISIE PRECEDENTE
65026 VOIT$ : NOM DE LA VOITURE
65027 Z : UNITE DE COMPTAGE
65028 -----FIN DE LISTING-----

```